Lecture 4.2

A Simple CGE Model for GAMS

David Roland-Holst, Sam Heft-Neal, and Anaspree Chaiwan
UC Berkeley and Chiang Mai University

Training Workshop
Economywide Assessment of High Impact Animal Disease
14-18 January 2013
InterContinental Hotel, Phnom Penh, Cambodia

Introduction

- The Generalized Algebraic Modeling System (GAMS) is a high-level programming language to specify and implement CGE models.
- GAMS is relatively easy to learn for a computer-literate individual with some knowledge of linear algebra.
- It is also flexible enough to implement very large models on a PC platform.
- Here we introduce the GAMS language with a practical example, i.e. specification and calibration of a simple CGE specification.

A Simple CGE Model

- One of the first computable general equilibrium (CGE) models was that of Johansen (1960).
- A Johansen-style, or log-linear CGE model is written as a system of equations linear in proportional changes of the variables.
- Perhaps the best-known analytical statement of this type of model was given by Jones (1965).
- We first set out the Jones algebra and then describe its translation into the GAMS language.

Simple CGE: The Jones Algebra

- Consider a two-sector model with the following production structure $Y_j = F_j(L_j, K_j)$, where j=1,2, the first sector produces importable goods (Y_1), and the second produces exportable goods (Y_2).
- The factors are defined by L_j, labor input into sector-j production, with $L_1 + L_2 = L$, where L is the employment level, and K_j, capital input into sector-j production, and $K_1 + K_2 = K$, where K is the current stock of capital.

Notation

In order to formulate a complete model, more notation is needed:

- Let w denote the wage rate, r the capital rental rate.
- Now let p_j and pw_j denote the domestic and world prices of good j, respectively, while a_{ij} is the input coefficient for input i into the production of good j.
- Finally, let t_1 denote an import tariff and s_2 is an export subsidy.

Model Specification

- This notation and the assumptions of constant returns to scale in production and perfect competition yield the following general equilibrium system:

Fixed Employment Conditions

$$a_{L1}Y_1 + a_{L2}Y_2 = L \qquad (6.1)$$

$$a_{K1}Y_1 + a_{K2}Y_2 = K \qquad (6.2)$$

Average Cost Pricing

$$wa_{L1} + ra_{K1} = p_1 \qquad (6.3)$$

$$wa_{L2} + ra_{K2} = p_2 \qquad (6.4)$$

Conditional input coefficient functions:

$$a_{L1} = a_{L1}(w,r) \quad (6.5)$$

$$a_{L2} = a_{L2}(w,r) \quad (6.6)$$

$$a_{K1} = a_{K1}(w,r) \quad (6.7)$$

$$a_{K2} = a_{K2}(w,r) \quad (6.8)$$

Domestic Price Equations

$$p_1 = (1+t_1)p_{w1} \qquad (6.9)$$

$$p_2 = (1+s_2)p_{w2} \qquad (6.10)$$

In Log-linear or Percent Change Form

$$\lambda_{L1} Y_1 + \lambda_{L2} Y_2 = L - \lambda_{L1}\hat{a}_{L1} - \lambda_{L2}\hat{a}_{L2} \quad (6.11)$$

$$\lambda_{K1} Y_1 + \lambda_{K2} Y_2 = K - \lambda_{K1}\hat{a}_{K1} - \lambda_{K2}\hat{a}_{K2} \quad (6.12)$$

$$\theta_{L1}\hat{w} + \theta_{K1}\hat{r} = \hat{p}_1 \quad (6.13)$$

$$\theta_{L2}\hat{w} + \theta_{K2}\hat{r} = \hat{p}_2 \quad (6.14)$$

$$\hat{a}_{L1} = \theta_{K1}\sigma_1(\hat{r} - \hat{w}) \quad (6.15)$$

$$\hat{a}_{L2} = \theta_{K2}\sigma_2(\hat{r} - \hat{w}) \quad (6.16)$$

$$\hat{a}_{K1} = \theta_{L1}\sigma_1(\hat{w} - \hat{r}) \quad (6.17)$$

$$\hat{a}_{K2} = \theta_{L2}\sigma_2(\hat{w} - \hat{r}) \quad (6.18)$$

$$\hat{p}_1 = \hat{p}_{w1} + dt_1/(1+t_1) \quad (6.19)$$

$$\hat{p}_2 = \hat{p}_{w2} + ds_2/(1+s_2) \quad (6.20)$$

Elements of GAMS

1. Operations
2. Relations
3. Syntax
4. Model Structure

Operations

**	exponentiation
* /	multiplications and division
+ -	addition and subtraction

Relations

lt, le, eq, ne, ge, gt	less than, less than or equal, not equal, etc.
not	not
and	and
or xor	or, either or

Syntax

GAMS programs consist of a series of statements followed by semicolons:

```
Statement ;
.
.
.
Statement ;
```

Model Structure

GAMS models are commonly structured as follows:

Data:

- SAM
- Parameters and other data

Definitions:

- Sets
- Parameters
- Initial values
- Variables
- Equations

Model:

Solution:

Display:

Definitions

Definitions in GAMS generally take the form:

```
sets
   setname name1   text  /elements/
        .
        .
        .
   setname namen     text  /elements/
        ;
```

Which for the Jones Model would look like:

sets

```
        i         industries  / 1*2 /
        f         factors / L,K /
   ;
```

Data I

Data can be introduced as scalars or parameters, e.g.

```
scalars
        dummy  named  / 1.0 /
   ;
parameters
        parameter  name1  text  /values/
                ...
        parameter namen  text  /values/
   ;
```

Data II

Which in the present application would be:

```
parameters
        lambda(f,i)      factor allocation
        theta(f,i)       factor income share
        sigma(i)         elasticity of factor substitution
                 / 1  0.8
                   2  0.9 /
        t(i)             initial tariff
        s(i)             initial subsidy
        dt(i)            change in tariff
        tarhat(i)        proportional change in tariff
        subhat(i)        proportional change in export subsidy
        cphat(i)         proportional change in price due to commercial
                         policy
        ;
```

Data III

Parameter value assignment can be made with equations:

```
t('1') = 0.20 ;
t('2') = 0.30 ;
dt('1') = 0.10 ;
dt('2') = 0.15 ;

tarhat(i) $ ( t(i)  gt  0 )  =  dt(i)/(1 + t(i)) ;
subhat(i) $ ( s(i) gt 0 )  =  ds(i)/(1 + s(i)) ;

cphat(i)  =  tarhat(i) $ t(i) + subhat(i) $ s(i) ;
```

Data IV

or table assignments:

```
table  lambda(f,i)

                1       2
        L       0.50    0.50
        K       0.25    0.75
    ;

  table theta(f,i)
                1       2
        L       0.60    0.40
        K       0.40    0.60
    ;
```

Variable Definition

```
variables
        yhat(i)         proportional change in production
        ahat(f,i)       proportional change in input
        what            proportional change in wage rate
        rhat            proportional change in capital rental rate
        phat(i)         proportional change in domestic price

        lhat            proportional change in labor endowment
        khat            proportional change in capital endowment
        psthat(i)       proportional change in world price

        omega           dummy variable for objective function
        ;
```

Equation Definition

```
equations
        fxelab      fixed employment of labor
        fxecap      fixed employment of capital
        acp(i)      average cost pricing
        linp(i)     labor input
        kinp(i)     capital input
        domp(i)     domestic prices
        obj         objective
        ;
```

Equation Specification

```
fxelab..    sum(i, lambda('l',i)*yhat(i)) =e= lhat -  sum(i,
lambda('l',i)*ahat('l',i));

fxecap..    sum(i, lambda('k',i)*yhat(i)) =e= khat - sum(i,
lambda('k',i)*ahat('k',i));

acp(i)..    theta('l',i)*what + theta('k',i)*rhat =e= phat(i);

linp(i)..   ahat('l',i) =e= theta('k',i)*sigma(i)*(rhat-what);

kinp(i)..   ahat('k',i) =e= theta('l',i)*sigma(i)*(what-rhat);

domp(i)..   phat(i) =e= psthat(i) + cphat(i);

obj..       omega =e= dummy;
```

Variable Initialization

Variables can be free or fixed

.fx fixed value

.lo the lower bound

.up the upper bound

.l the activity level

.m the marginal value

```
lhat.fx = 0.00 ;
khat.fx = 0.00;
psthat.fx('1') = 0.00;
psthat.fx('2') = 0.00;
```

Model Definition, Solution, and Output

```
model simple /all/;

   solve simple maximizing omega using nlp;

   display yhat.l, ahat.l, what.l, rhat.l, phat.l, lhat.l,
        khat.l, psthat.l;
```

Questions?